5. Három fajt tartalmazó modellek

Többfajos modellekkel részletesebben L. A. Buttel, R. Durret és S. A. Levin [2] foglalkozott és vizsgált az évezredforduló elején.

5.1. Szimmetrikus preemptív versengés három fajra

Ebben a fejezetben a szimmetrikus preemptív versengés háromfajos modelljét tárgyaljuk. Ebben a modellben a fajok között nincs interakció, a három faj békésen él egymás mellett, és úgy versengenek, hogy csak az üres területeket képesek elfoglalni, egymást nem kolonizálják. A modellt a teljesség kedvéért tekintjük, mivel a kétfajos változat alapján nyilvánvaló hogy a három fajból legalább kettő mindig kihal.

5.1.1. A modell

A szimmetrikus preemptív versengés modelljében mindegyik faj kolonizációjának sebessége arányos az általa elfoglalt és az üres foltok arányának szorzatával. A halálozás ebben a modellben is a faj belső jellemzője, ezért csak az adott faj elfoglalt foltok arányával arányos. Legyen $p_1(t)$ az első, $p_2(t)$ a második, $p_3(t)$ a harmadik ($0 \le p_1(t) + p_2(t) + p_3(t) \le 1$) faj által elfoglalt terület aránya a t időpillanatban. A modellt az alábbi egyenletrendszer írja le:

$$\begin{aligned} p_1'(t) &= k_1\, p_1(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) - e_1\, p_1(t) \\ p_2'(t) &= k_2\, p_2(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) - e_2\, p_2(t) \\ p_3'(t) &= k_3\, p_3(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) - e_3\, p_3(t) \end{aligned} \tag{5.1.1}$$

ahol k_1, k_2, k_3 $(0 \le k_i,\ i = 1, 2, 3)$ konstansok a globális kolonizációs ráták, e_1, e_2, e_3 $(0 \le e_i,\ i = 1, 2, 3)$ konstansok a globális halálozási ráták.

A továbbiakban az egyszerűség kedvéért használjuk a következő jelölést: $p_i := p_i(t),\ i = 1, 2, 3$. Az alábbi ábra mutatja az egyes fajok által elfoglalt és üres foltok közti változást.



5.1.1. ábra

A szimmetrikus preemptív versengés három fajra; e_i a kihalási ráta, k_i a kolonizációs ráta, p_i az elfoglalt foltok, L a lakatlan (üres) folt ($i = 1,\ 2,\ 3$)

5.1.2. Az egyensúlyi helyzetek

Változók és paraméterek:

```
Clear[e1, e2, e3, k1, k2, k3];
var = {p1, p2, p3};
parm = {k1, k2, k3, e1, e2, e3};
```

Az (5.1.1) egyenletrendszer jobb oldala:

```
Levins3 = {p1 * (k1 * (1 - p1 - p2 - p3) - e1),
   p2 * (k2 * (1 - p1 - p2 - p3) - e2), p3 * (k3 * (1 - p1 - p2 - p3) - e3)};
```

Az (5.1.1) egyenletrendszer egyensúlyi helyzetei:

```
Levins3Egyensuly = Solve[Levins3 == {0, 0, 0}, var];
Levins3Egyensuly // Column
```

$$\begin{array}{l} \{p1 \to 0,\ p2 \to 0,\ p3 \to 0\} \\ \left\{p1 \to -\frac{e1-k1}{k1},\ p2 \to 0,\ p3 \to 0\right\} \\ \left\{p2 \to -\frac{e2-k2}{k2},\ p1 \to 0,\ p3 \to 0\right\} \\ \left\{p3 \to -\frac{e3-k3}{k3},\ p1 \to 0,\ p2 \to 0\right\} \end{array}$$

Az egyensúlyi helyzeteket látva nyilvánvaló az alábbi tétel, mely a kétfajos neutrális modell esetén is igaz volt.

5.1.1. TÉTEL

Nincs a rendszernek olyan egyensúlyi helyzete, ahol a legalább két faj által elfoglalt terület pozitív lenne.

5.1.1. MEGJEGYZÉS

Mivel $0 \le p_1(t) + p_2(t) + p_3(t) \le 1$, azonnal látható, hogy a nemtriviális egyensúlyi helyzetek pozitívak, ha $\frac{e_i}{k_i} \le 1$, $i = 1, 2, 3$. (Az egyfajos Levins-modell és a kétfajos neutrális modell esetén ugyanez volt a feltétel.)

5.1.3. Az egyensúlyi helyzetek stabilitási tulajdonságai

Az (5.1.1) egyenletrendszer jobb oldalának Jacobi mátrixai:

```
Levins3Jacobi = FullSimplify[D[Levins3, {var}]]
```

```
{{-e1 - k1 (-1 + 2 p1 + p2 + p3), -k1 p1, -k1 p1},
 {-k2 p2, -e2 - k2 (-1 + p1 + 2 p2 + p3), -k2 p2},
 {-k3 p3, -k3 p3, -e3 - k3 (-1 + p1 + p2 + 2 p3)}}
```

```
linLevins3 = Levins3Jacobi /. Levins3Egyensuly;
```

```
Map[MatrixForm, linLevins3] // Column
```

$$\begin{pmatrix} -e1+k1 & 0 & 0 \\ 0 & -e2+k2 & 0 \\ 0 & 0 & -e3+k3 \end{pmatrix}$$

$$\begin{pmatrix} -e1-\left(-1-\frac{2\,(e1-k1)}{k1}\right) k1 & e1-k1 & e1-k1 \\ 0 & -e2-\left(-1-\frac{e1-k1}{k1}\right) k2 & 0 \\ 0 & 0 & -e3-\left(-1-\frac{e1-k1}{k1}\right) k3 \end{pmatrix}$$

$$\begin{pmatrix} -e1-k1\left(-1-\frac{e2-k2}{k2}\right) & 0 & 0 \\ e2-k2 & -e2-\left(-1-\frac{2\,(e2-k2)}{k2}\right) k2 & e2-k2 \\ 0 & 0 & -e3-\left(-1-\frac{e2-k2}{k2}\right) k3 \end{pmatrix}$$

$$\begin{pmatrix} -e1-k1\left(-1-\frac{e3-k3}{k3}\right) & 0 & 0 \\ 0 & -e2-k2\left(-1-\frac{e3-k3}{k3}\right) & 0 \\ e3-k3 & e3-k3 & -e3-\left(-1-\frac{2\,(e3-k3)}{k3}\right) k3 \end{pmatrix}$$

A Jacobi mátrixok sajátértékei rendre:

```
ev = Map[MatrixForm,
  Map[FullSimplify[Eigenvalues[#]] &, linLevins3]]
```

$$\left\{\begin{pmatrix} -e1+k1 \\ -e2+k2 \\ -e3+k3 \end{pmatrix}, \begin{pmatrix} e1-k1 \\ -e2+\frac{e1\,k2}{k1} \\ -e3+\frac{e1\,k3}{k1} \end{pmatrix}, \begin{pmatrix} -e1+\frac{e2\,k1}{k2} \\ e2-k2 \\ -e3+\frac{e2\,k3}{k2} \end{pmatrix}, \begin{pmatrix} -e1+\frac{e3\,k1}{k3} \\ -e2+\frac{e3\,k2}{k3} \\ e3-k3 \end{pmatrix}\right\}$$

5.1.2. TÉTEL

(i) Ha $1<\frac{e_i}{k_i}$, $i=1,2,3$, akkor a $(0,\ 0,\ 0)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,i$) Ha $\frac{e_1}{k_1}<1$, $\frac{e_1}{k_1}<\frac{e_2}{k_2}$ és $\frac{e_1}{k_1}<\frac{e_3}{k_3}$, akkor a $\left(1-\frac{e_1}{k_1},\ 0,\ 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,i\,i$) Ha $\frac{e_2}{k_2}<1$, $\frac{e_2}{k_2}<\frac{e_1}{k_1}$ és $\frac{e_2}{k_2}<\frac{e_3}{k_3}$, akkor a $\left(0,\ 1-\frac{e_2}{k_2},\ 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,v$) Ha $\frac{e_3}{k_3}<1$, $\frac{e_3}{k_3}<\frac{e_1}{k_1}$ és $\frac{e_3}{k_3}<\frac{e_2}{k_2}$, akkor a $\left(0,\ 0,\ 1-\frac{e_3}{k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

BIZONYÍTÁS

A következő egyenlőtlenségeknek kell teljesülniük:

(i) $\quad -e_i+k_i<0 \Leftrightarrow k_i<e_i \Leftrightarrow 1<\frac{e_i}{k_i}$, $i=1,2$

($i\,i$) $\quad e_1-k_1<0 \Leftrightarrow e_1<k_1 \Leftrightarrow \frac{e_1}{k_1}<1$ és

$$-e_j+\frac{e_1\,k_j}{k_1}<0 \Leftrightarrow \frac{e_1\,k_j}{k_1}<e_j \Leftrightarrow \frac{e_1}{k_1}<\frac{e_j}{k_j},\ j=2,3$$

$(i\,i\,i)$ és $(i\,v)$ a $(i\,i)$ bizonyításának analógiája.

■

5.1.4. Grafikus megjelenítés - interaktív vizsgálatok

Lehetőség van a rendszer interaktív vizsgálatára. Kézzel változtathatjuk a paraméterek értékeit, és láthatjuk hogyan változik az iránymező és az adott kezdeti értékekhez tartozó megoldás.



5.2. Hierarchikus felülkolonizációs modell

Ebben a fejezetben olyan aszimmetrikus modellt vizsgálunk, ahol – a kétfajos hierarchikus felülkolonizációs modellhez hasonlóan – a fajok között hierarchia áll fenn. Minden megszorítás nélkül feltehető, hogy a $p_1(t)$-vel jelölt faj a legdominánsabb, a hierarchiában őt a $p_2(t)$-vel jelölt faj követi, majd a leggyengébb $p_3(t)$.

5.2.1. A modell

Az alábbi hierarchikus felülkolonizációs modellben egy-egy faj kolonizációjának sebessége arányos az általa elfoglalt és azon foltok arányának szorzatával, amelyet képes kolonizálni. A halálozás a faj belső jellemzője, ezért csak az adott faj által elfoglalt foltok arányával arányos. Legyen $p_1(t)$ az első, $p_2(t)$ a második, $p_3(t)$ a harmadik ($0 \le p_1(t) + p_2(t) + p_3(t) \le 1$) faj által elfoglalt terület aránya a t időpillanatban.

Ebben a modellben az első faj képes elfoglalni minden általa még nem elfoglalt foltot, a második gyengébb nála, és nem képes betörni az első faj foltjaira, de el tudja foglalni az üres és a harmadik faj által elfoglalt foltokat. Végül a harmadik faj csak üres foltokat képes kolonizálni. A kolonizációs képességek nem függnek az elfoglalandó foltok típusától. A modellt az alábbi egyenletrendszer írja le:

$$\begin{aligned} p_1'(t) &= k_1\, p_1(t)\,(1 - p_1(t)) - e_1\, p_1(t) \\ p_2'(t) &= k_2\, p_2(t)\,(1 - p_1(t) - p_2(t)) - k_1\, p_1(t)\, p_2(t) - e_2\, p_2(t) \\ p_3'(t) &= k_3\, p_3(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) - k_1\, p_1(t)\, p_3(t) - k_2\, p_2(t)\, p_3(t) - e_3\, p_3(t) \end{aligned} \tag{5.2.2}$$

ahol k_1, k_2, k_3 ($0 \le k_i,\ i = 1, 2, 3$) konstansok a globális kolonizációs ráták, e_1, e_2, e_3 ($0 \le e_i,\ i = 1, 2, 3$) konstansok a globális halálozási ráták. Az alábbi ábra mutatja az egyes fajok által elfoglalt és üres foltok közti változást.



5.2.1. ábra

Hierarchikus felülkolonizációs modell három fajra; e_i a kihalási ráta, k_i a kolonizációs ráta, p_i az elfoglalt folt, L a lakatlan (üres) folt ($i = 1,\ 2,\ 3$)

Az (5.2.2) modellben lényeges egyszerűsítés, hogy nem vesszük figyelembe a kolonizálandó foltokon élő faj ellenállását a betolakodóval szemben. A továbbiakban az egyszerűség kedvéért használjuk a következő jelölést: $p_i := p_i(t),\ i = 1, 2, 3$.

5.2.2. Az egyensúlyi helyzetek

Változók és paraméterek:

```
Clear[e1, e2, e3, k1, k2, k3];
var = {p1, p2, p3};
parm = {k1, k2, k3, e1, e2, e3};
```

Az (5.2.2) egyenletrendszer jobb oldala:

```
AszFel3 =
  {p1 * (k1 * (1 - p1) - e1),
   p2 * (k2 * (1 - p1 - p2) - e2 - k1 p1),
   p3 * (k3 * (1 - p1 - p2 - p3) - e3 - k1 p1 - k2 p2)};
```

Az (5.2.2) egyenletrendszer egyensúlyi helyzetei:

```
AszFel3Egyensuly =
  FullSimplify[Solve[AszFel3 == {0, 0, 0}, var]];
AszFel3Egyensuly // Column
```

$$\{p1 \to 0,\ p2 \to 0,\ p3 \to 0\}$$
$$\left\{p1 \to 1 - \frac{e1}{k1},\ p2 \to 0,\ p3 \to 0\right\}$$
$$\left\{p2 \to 1 - \frac{e2}{k2},\ p3 \to 0,\ p1 \to 0\right\}$$
$$\left\{p2 \to \frac{-k1\ (e2+k1)+e1\ (k1+k2)}{k1\ k2},\ p3 \to 0,\ p1 \to 1 - \frac{e1}{k1}\right\}$$
$$\left\{p3 \to 1 - \frac{e3}{k3},\ p1 \to 0,\ p2 \to 0\right\}$$
$$\left\{p3 \to \frac{-k1\ (e3+k1)+e1\ (k1+k3)}{k1\ k3},\ p1 \to 1 - \frac{e1}{k1},\ p2 \to 0\right\}$$
$$\left\{p3 \to \frac{-k2\ (e3+k2)+e2\ (k2+k3)}{k2\ k3},\ p2 \to 1 - \frac{e2}{k2},\ p1 \to 0\right\}$$
$$\left\{p3 \to \frac{-e1+e2+k1}{k2} + \frac{e2\ k1-e3\ k1-e1\ k2}{k1\ k3},\ p2 \to \frac{-k1\ (e2+k1)+e1\ (k1+k2)}{k1\ k2},\ p1 \to 1 - \frac{e1}{k1}\right\}$$

Az egyenletrendszernek van együttélő egyensúlyi helyzete, ez a fajok közti dominanciának köszönhető. Fontos megjegyezni, hogy $0 \le p_1(\mathrm{t})+p_2(\mathrm{t})+p_3(t) \le 1$. Így a fentiek pontosan akkor egyensúlyi helyzetek, ha az egyenlőtlenség igaz rájuk a megfelelő behelyettesítéssel. Ha csak az egyik faj képes a túlélésre, akkor teljesülnie kell a $0 \le 1 - \frac{e_i}{k_i} \Leftrightarrow \frac{e_i}{k_i} \le 1,\ i = 1, 2$ egyenlőtlenségnek ugyanúgy, mint a neutrális modell esetén. A többi egyensúlyi helyzetre ezt a feltételt az alábbi tétel bizonyításában vizsgáljuk.

5.2.3. Az egyensúlyi helyzetek stabilitási tulajdonságai

Az (5.2.2) egyenletrendszer jobb oldalának Jacobi mátrixai:

```
AszFel3Jacobi = FullSimplify[D[AszFel3, {var}]]
```

```
{{-e1 + k1 - 2 k1 p1, 0, 0},
 {-(k1 + k2) p2, -e2 + k2 - k1 p1 - k2 p1 - 2 k2 p2, 0}, {-(k1 + k3) p3,
  -(k2 + k3) p3, -e3 + k3 - k1 p1 - k3 p1 - k2 p2 - k3 p2 - 2 k3 p3}}
```

```
linAszFel3 = AszFel3Jacobi /. AszFel3Egyensuly;
Map[MatrixForm, linAszFel3] // Column
```

A Jacobi mátrixok sajátértékei rendre:

```
ev = Map[MatrixForm,
  Map[FullSimplify[Eigenvalues[#]] &, linAszFel3]]
```

$$\left\{\begin{pmatrix}-e1+k1\\ -e2+k2\\ -e3+k3\end{pmatrix}, \begin{pmatrix}e1-k1\\ e1-e2-k1+\frac{e1\,k2}{k1}\\ e1-e3-k1+\frac{e1\,k3}{k1}\end{pmatrix}, \begin{pmatrix}-e1+k1\\ e2-k2\\ e2-e3-k2+\frac{e2\,k3}{k2}\end{pmatrix},\right.$$

$$\begin{pmatrix}e1-k1\\ e2+k1-\frac{e1\,(k1+k2)}{k1}\\ e2-e3-\frac{e1\,k2}{k1}+\frac{(-e1+e2+k1)\,k3}{k2}\end{pmatrix}, \begin{pmatrix}-e1+k1\\ -e2+k2\\ e3-k3\end{pmatrix}, \begin{pmatrix}e1-k1\\ e1-e2-k1+\frac{e1\,k2}{k1}\\ e3+k1-\frac{e1\,(k1+k3)}{k1}\end{pmatrix},$$

$$\left.\begin{pmatrix}-e1+k1\\ e2-k2\\ e3+k2-\frac{e2\,(k2+k3)}{k2}\end{pmatrix}, \begin{pmatrix}e1-k1\\ e2+k1-\frac{e1\,(k1+k2)}{k1}\\ -e2+e3+\frac{e1\,k2}{k1}-\frac{(-e1+e2+k1)\,k3}{k2}\end{pmatrix}\right\}$$

5.2.1. TÉTEL

(i) Ha $1 < \frac{e_i}{k_i}$, $i = 1, 2, 3$, akkor a $p = (0, 0, 0)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,i$) Ha $\frac{e_1}{k_1} < 1$ és $\frac{e_1}{k_1} < \frac{e_j+k_1}{k_1+k_j}$, $j = 2, 3$, akkor a $p = \left(1 - \frac{e_1}{k_1}, 0, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,i\,i$) Ha $1 < \frac{e_1}{k_1}$, $\frac{e_2}{k_2} < 1$ és $\frac{e_2}{k_2} < \frac{e_3+k_2}{k_2+k_3}$, akkor a $p = \left(0, 1 - \frac{e_2}{k_2}, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,v$) Ha $1 < \frac{e_1}{k_1}$, $1 < \frac{e_2}{k_2}$ és $\frac{e_3}{k_3} < 1$, akkor a $p = \left(0, 0, 1 - \frac{e_3}{k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

BIZONYÍTÁS

A következő egyenlőtlenségeknek kell teljesülniük:

(i) $\quad -e_i + k_i < 0 \iff k_i < e_i \iff 1 < \frac{e_i}{k_i}$, $i = 1, 2, 3$

($i\,i$) $\quad e_1 - k_1 < 0 \iff e_1 < k_1 \iff \frac{e_1}{k_1} < 1$ és

$$e_1 - e_2 - k_1 + \frac{e_1 k_2}{k_1} < 0 \Leftrightarrow e_1 k_1 - e_2 k_1 - {k_1}^2 + e_1 k_2 < 0 \Leftrightarrow e_1 k_1 + e_1 k_2 < e_2 k_1 + {k_1}^2 \Leftrightarrow$$

$$\Leftrightarrow e_1(k_1 + k_2) < k_1(e_2 + k_1) \Leftrightarrow \frac{e_1}{k_1} < \frac{e_2+k_1}{k_1+k_2}$$ és ugyanígy e_3 és k_3 esetén

(*i i i*) $-e_1 + k_1 < 0 \Leftrightarrow k_1 < e_1 \Leftrightarrow 1 < \frac{e_1}{k_1}$ és

$e_2 - k_2 < 0 \Leftrightarrow e_2 < k_2 \Leftrightarrow \frac{e_2}{k_2} < 1$ és

$$e_2 - e_3 - k_2 + \frac{e_2 k_3}{k_2} < 0 \Leftrightarrow e_2 k_2 - e_3 k_2 - {k_2}^2 + e_2 k_3 < 0 \Leftrightarrow e_2 k_2 + e_2 k_3 < e_3 k_2 + {k_2}^2 \Leftrightarrow$$

$$\Leftrightarrow e_2 (k_2 + k_3) < k_2 (e_3 + k_2) \Leftrightarrow \frac{e_2}{k_2} < \frac{e_3+k_2}{k_2+k_3}$$

(*i v*) $-e_1 + k_1 < 0 \Leftrightarrow k_1 < e_1 \Leftrightarrow 1 < \frac{e_1}{k_1}$ és

$-e_2 + k_2 < 0 \Leftrightarrow k_2 < e_2 \Leftrightarrow 1 < \frac{e_2}{k_2}$ és

$e_3 - k_3 < 0 \Leftrightarrow e_3 < k_3 \Leftrightarrow \frac{e_3}{k_3} < 1.$

■

5.2.2. TÉTEL

(*i*) Az $\left(1 - \frac{e_1}{k_1}, \frac{-k_1(e_2+k_1)+e_1(k_1+k_2)}{k_1 k_2}, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilitásának szükséges feltétele: $\frac{e_1}{k_1} < 1$, $\frac{e_2+k_1}{k_1+k_2} < \frac{e_1}{k_1}$ és $e_2 - k_2 \le e_1 - k_1 \le e_2$.

(*i i*) Az $\left(1 - \frac{e_1}{k_1}, 0, \frac{-k_1(e_3+k_1)+e_1(k_1+k_3)}{k_1 k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilis, ha $\frac{e_1}{k_1} < 1$, $\frac{e_1}{k_1} < \frac{e_2+k_1}{k_1+k_2}$ és $\frac{e_3+k_1}{k_1+k_3} < \frac{e_1}{k_1}$.

(*i i i*) Ha $1 < \frac{e_1}{k_1}$, $\frac{e_2}{k_2} < 1$ és $\frac{e_3+k_2}{k_2+k_3} < \frac{e_2}{k_2}$, akkor a $p = \left(0, 1 - \frac{e_2}{k_2}, \frac{-k_2(e_3+k_2)+e_2(k_2+k_3)}{k_2 k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

(*i v*) Az $\left(1 - \frac{e_1}{k_1}, \frac{-k_1(e_2+k_1)+e_1(k_1+k_2)}{k_1 k_2}, \frac{-e_1+e_2+k_1}{k_2} + \frac{k_1 e_2 - e_3 k_1 - e_1 k_2}{k_1 k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilitásának szükséges feltétele: $\frac{e_2+k_1}{k_1+k_2} < \frac{e_1}{k_1}$, $1 < \frac{e_1}{k_1}$, $\frac{e_1}{k_1} \le \frac{e_2-e_3+k_3}{k_2}$ és $\frac{e_2-e_3}{k_2} \le \frac{e_1}{k_1}$.

BIZONYÍTÁS

A következő egyenlőtlenségeknek kell teljesülniük:

(*i*) $e_1 - k_1 < 0 \Leftrightarrow e_1 < k_1 \Leftrightarrow \frac{e_1}{k_1} < 1$ és

$$e_2 + k_1 - \frac{e_1(k_1+k_2)}{k_1} < 0 \Leftrightarrow e_2 k_1 + {k_1}^2 - e_1 k_1 - e_1 k_2 < 0 \Leftrightarrow e_2 k_1 + {k_1}^2 < e_1 k_1 + e_1 k_2 \Leftrightarrow$$

$$\Leftrightarrow k_1 (e_2 + k_1) < e_1 (k_1 + k_2) \Leftrightarrow \frac{e_2+k_1}{k_1+k_2} < \frac{e_1}{k_1},$$

valamint szükség lenne továbbá $e_2 - e_3 - \frac{e_1 k_2}{k_1} + \frac{(-e_1+e_2+k_1) k_3}{k_2} < 0$ teljesülésére, viszont ennek kiszámolása még a számítógépnek is problémát okozott. Vizsgáljuk meg erre az egyensúlyi helyzetre a $0 \le p_1(t)+p_2(t)+p_3(t) \le 1$ feltételt:

$$1 - \frac{e_1}{k_1} + \frac{-k_1(e_2+k_1)+e_1(k_1+k_2)}{k_1 k_2} = \frac{e_1-e_2-k_1+k_2}{k_2} \ge 0 \Leftrightarrow e_1 - k_1 \ge e_2 - k_2$$ és

$$\frac{e_1-e_2-k_1+k_2}{k_2} \le 1 \Leftrightarrow e_1 - e_2 - k_1 + k_2 \le k_2 \Leftrightarrow e_1 - k_1 \le e_2$$

(*i i*) $e_1 - k_1 < 0 \Leftrightarrow e_1 < k_1 \Leftrightarrow \frac{e_1}{k_1} < 1$ és

$$e_1 - e_2 - k_1 + \frac{e_1 k_2}{k_1} < 0 \Leftrightarrow e_1 k_1 - e_2 k_1 - k_1^2 + e_1 k_2 < 0 \Leftrightarrow e_1 k_1 + e_1 k_2 < e_2 k_1 + k_1^2 \Leftrightarrow$$

$$\Leftrightarrow e_1(k_1 + k_2) < k_1(e_2 + k_1) \Leftrightarrow \frac{e_1}{k_1} < \frac{e_2+k_1}{k_1+k_2} \text{ és}$$

$$e_3 + k_1 - \frac{e_1(k_1+k_3)}{k_1} < 0 \Leftrightarrow e_3 k_1 + k_1^2 - e_1 k_1 - e_1 k_3 < 0 \Leftrightarrow k_1^2 + e_3 k_1 < e_1 k_1 + e_1 k_3 \Leftrightarrow$$

$$\Leftrightarrow k_1(e_3 + k_1) < e_1(k_1 + k_3) \Leftrightarrow \frac{e_3+k_1}{k_1+k_3} < \frac{e_1}{k_1}$$

(*i i i*) $-e_1 + k_1 < 0 \Leftrightarrow k_1 < e_1 \Leftrightarrow 1 < \frac{e_1}{k_1}$ és

$$e_2 - k_2 < 0 \Leftrightarrow e_2 < k_2 \Leftrightarrow \frac{e_2}{k_2} < 1 \text{ és}$$

$$e_3 + k_2 - \frac{e_2(k_2+k_3)}{k_2} < 0 \Leftrightarrow e_3 k_2 + k_2^2 - e_2 k_2 - e_2 k_3 < 0 \Leftrightarrow e_3 k_2 + k_2^2 < e_2 k_2 + e_2 k_3 \Leftrightarrow$$

$$\Leftrightarrow k_2(e_3 + k_2) < e_2(k_2 + k_3) \Leftrightarrow \frac{e_3+k_2}{k_2+k_3} < \frac{e_2}{k_2}$$

(*i v*) $-e_1 + k_1 < 0 \Leftrightarrow k_1 < e_1 \Leftrightarrow 1 < \frac{e_1}{k_1}$ és

$$e_2 + k_1 - \frac{e_1(k_1+k_2)}{k_1} < 0 \Leftrightarrow e_2 k_1 + k_1^2 - e_1 k_1 - e_1 k_2 < 0 \Leftrightarrow e_2 k_1 + k_1^2 < e_1 k_1 + e_1 k_2 \Leftrightarrow$$

$$\Leftrightarrow k_1(e_2 + k_1) < e_1(k_1 + k_2) \Leftrightarrow \frac{e_2+k_1}{k_1+k_2} < \frac{e_1}{k_1},$$

valamint szükség lenne továbbá a $-e_2 + e_3 + \frac{e_1 k_2}{k_1} - \frac{(-e_1+e_2+k_1)k_3}{k_2} < 0$ teljesülésére, viszont ennek kiszámolása szintén problémát okozott még a számítógépnek is. Vizsgáljuk meg erre az egyensúlyi helyzetre a $0 \le p_1(t)+p_2(t)+p_3(t) \le 1$ feltételt:

$$\texttt{Simplify}\left[\frac{-e1 + e2 + k1}{k2} + \frac{e2\,k1 - e3\,k1 - e1\,k2}{k1\,k3} + \frac{-k1\,(e2 + k1) + e1\,(k1 + k2)}{k1\,k2} + 1 - \frac{e1}{k1}\right]$$

$$\frac{e2\,k1 - e3\,k1 - e1\,k2 + k1\,k3}{k1\,k3}$$

Így

$$\frac{e_2 k_1 - e_3 k_1 - e_1 k_2 + k_1 k_3}{k_1 k_3} \ge 0 \Leftrightarrow e_2 k_1 - e_3 k_1 - e_1 k_2 + k_1 k_3 \ge 0 \Leftrightarrow e_2 k_1 - e_3 k_1 + k_1 k_3 \ge e_1 k_2 \Leftrightarrow$$

$$\Leftrightarrow k_1(e_2 - e_3 + k_3) \ge e_1 k_2 \Leftrightarrow \frac{e_1}{k_1} \le \frac{e_2 - e_3 + k_3}{k_2} \text{ és}$$

$$\frac{e_2 k_1 - e_3 k_1 - e_1 k_2 + k_1 k_3}{k_1 k_3} \le 1 \Leftrightarrow e_2 k_1 - e_3 k_1 - e_1 k_2 \le 0 \Leftrightarrow k_1(e_2 - e_3) \le e_1 k_2 \Leftrightarrow \frac{e_1}{k_1} \ge \frac{e_2 - e_3}{k_2}.$$

■

5.2.4. Grafikus megjelenítés - interaktív vizsgálatok

Lehetőség van a rendszer interaktív vizsgálatára. Kézzel változtathatjuk a paraméterek értékeit, és láthatjuk hogyan változik az iránymező és az adott kezdeti értékekhez tartozó megoldás.



5.3. Általános felülkolonizációs modell

Ebben a fejezetben olyan háromfajos modelleket vizsgálunk, amelyekben a fajok között interakciót vélünk felfedezni, de a fajok között nincs hierarchia, mindegyik faj képes kolonizálni és felülkolonizálni. Ez az úgynevezett szimmetrikus felülkolonizációs modell, ahol a kolonizációs ráta nem egyezik meg a felülkolonizációs rátával.

5.3.1. A modell

Az általános felülkolonizációs modellben mindhárom faj kolonizációjának sebessége arányos az elfoglalt és az el nem foglalt területek arányának szorzatával. A felülkolonizáció sebessége arányos az adott faj által elfoglalt és egy másik faj által elfoglalt terület arányának szorzatával. A halálozás a faj belső jellemzője, ezért csak az adott faj által elfoglalt foltok arányával arányos. Legyen $p_1(t)$ az első, $p_2(t)$ a második, $p_3(t)$ a harmadik ($0 \le p_1(t) + p_2(t) + p_3(t) \le 1$) faj által elfoglalt terület aránya a t időpillanatban. A modellt az alábbi egyenletrendszer írja le:

$$\begin{gathered}
p_1'(t) = \\
k_1\, p_1(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) + (c_1 - c_2)\, p_1(t)\, p_2(t) + (c_1 - c_3)\, p_1(t)\, p_3(t) - e_1\, p_1(t) \\
p_2'(t) = k_2\, p_2(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) + \\
(-c_1 + c_2)\, p_1(t)\, p_2(t) + (c_2 - c_3)\, p_2(t)\, p_3(t) - e_2\, p_2(t) \\
p_3'(t) = k_3\, p_3(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) + \\
(-c_1 + c_3)\, p_1(t)\, p_3(t) + (-c_2 + c_3)\, p_2(t)\, p_3(t) - e_3\, p_3(t)
\end{gathered} \tag{5.3.3}$$

ahol k_1, k_2, k_3 $(0 \le k_i,\ i = 1, 2, 3)$ konstansok a globális kolonizációs ráták, e_1, e_2, e_3 $(0 \le e_i,\ i = 1, 2, 3)$ konstansok a globális halálozási ráták, c_1, c_2, c_3 $(0 \le c_i,\ i = 1, 2, 3)$ konstansok a globális felülkolonizációs ráták. Az alábbi ábra mutatja az egyes fajok által elfoglalt és üres foltok közti változást.



5.3.1. ábra

Általános felülkolonizációs modell három fajra; e_i a kihalási ráta, k_i a kolonizációs ráta, p_i az elfoglalt folt, L a lakatlan (üres) folt ($i = 1,\ 2,\ 3$)

A továbbiakban az egyszerűség kedvéért használjuk a következő jelölést: $p_i := p_i(t),\ i = 1,\ 2,\ 3.$

5.3.2. Az egyensúlyi helyzetek

Változók és paraméterek:

```
Clear[k1, k2, k3, e1, e2, e3];
var = {p1, p2, p3};
parm = {k1, k2, k3, e1, e2, e3, c1, c2, c3};
```

Az (5.3.3) egyenletrendszer jobb oldala:

```
SzimFel31 = {p1 (k1 (1 - p1 - p2 - p3) - e1 + p2 (c1 - c2) + p3 (c1 - c3)),
             p2 (k2 (1 - p1 - p2 - p3) - e2 + p1 (-c1 + c2) + p3 (c2 - c3)),

   p3 (k3 (1 - p1 - p2 - p3) - e3 + p1 (-c1 + c3) + p2 (-c2 + c3))};
```

Az (5.3.3) egyenletrendszer egyensúlyi helyzetei:

```
SzimFel31Egyensuly =
  FullSimplify[Solve[SzimFel31 == {0, 0, 0}, var]];
SzimFel31Egyensuly // Column
```

$$\{p1 \to 0,\ p2 \to 0,\ p3 \to 0\}$$

$$\left\{p1 \to 1 - \frac{e1}{k1},\ p2 \to 0,\ p3 \to 0\right\}$$

$$\left\{p1 \to \frac{e2\,(c2+k1)-(c2+e1)\,k2+c1\,(-e2+k2)}{(c1-c2)\,(c1-c2-k1+k2)},\ p2 \to \frac{c1\,(e1-k1)-e2\,k1+c2\,(-e1+k1)+e1\,k2}{(c1-c2)\,(c1-c2-k1+k2)},\ p3 \to 0\right\}$$

$$\left\{p1 \to \frac{e3\,(c3+k1)-(c3+e1)\,k3+c1\,(-e3+k3)}{(c1-c3)\,(c1-c3-k1+k3)},\ p3 \to \frac{c1\,(e1-k1)-e3\,k1+c3\,(-e1+k1)+e1\,k3}{(c1-c3)\,(c1-c3-k1+k3)},\ p2 \to 0\right\}$$

$$\left\{p2 \to 1 - \frac{e2}{k2},\ p1 \to 0,\ p3 \to 0\right\}$$

$$\left\{p2 \to \frac{e3\,(c3+k2)-(c3+e2)\,k3+c2\,(-e3+k3)}{(c2-c3)\,(c2-c3-k2+k3)},\ p3 \to \frac{c2\,(e2-k2)-e3\,k2+c3\,(-e2+k2)+e2\,k3}{(c2-c3)\,(c2-c3-k2+k3)},\ p1 \to 0\right\}$$

$$\left\{p3 \to 1 - \frac{e3}{k3},\ p1 \to 0,\ p2 \to 0\right\}$$

Ennek az egyenletrendszernek nincs együttélő egyensúlyi helyzete. Fontos megjegyezni, hogy $0 \le p_1(t)+p_2(t)+p_3(t) \le 1$. Így a fentiek pontosan akkor egyensúlyi helyzetek, ha az egyenlőtlenség igaz rájuk a megfelelő behelyettesítéssel. Ha csak az egyik faj képes a túlélésre, akkor teljesülnie kell, hogy $0 \le 1 - \frac{e_i}{k_i} \Leftrightarrow \frac{e_i}{k_i} \le 1$, $i = 1, 2$, ugyanúgy, mint a neutrális modell esetén. A későbbiekben tárgyaljuk e feltétel teljesülését azon egyensúlyi helyzetekre, amikor két faj képes az együttélésre.

5.3.3. Az egyensúlyi helyzetek stabilitási tulajdonságai

Az (5.3.3) egyenletrendszer jobb oldalának Jacobi mátrixai:

```
SzimFel31Jacobi = FullSimplify[D[SzimFel31, {var}]]
        {{-e1 - c2 p2 - c3 p3 + c1 (p2 + p3) - k1 (-1 + 2 p1 + p2 + p3),
          (c1 - c2 - k1) p1, (c1 - c3 - k1) p1}, {- (c1 - c2 + k2) p2,
         -e2 - c1 p1 - c3 p3 + c2 (p1 + p3) - k2 (-1 + p1 + 2 p2 + p3),
         (c2 - c3 - k2) p2}, {- (c1 - c3 + k3) p3, - (c2 - c3 + k3) p3,
         -e3 - c1 p1 - c2 p2 + c3 (p1 + p2) - k3 (-1 + p1 + p2 + 2 p3)}}
linSzimFel31 = SzimFel31Jacobi /. SzimFel31Egyensuly;
Map[MatrixForm, linSzimFel31] // Column
```

A Jacobi mátrixok sajátértékei rendre:

```
ev = Map[FullSimplify[Eigenvalues[#]] &, linSzimFel31]
```

5.3.1. TÉTEL

(i) Ha $1 < \frac{e_i}{k_i}$, $i = 1, 2, 3$, akkor a $p = (0, 0, 0)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,i$) Ha $\frac{e_1}{k_1} < 1$, $\frac{e_1}{k_1} < \frac{c_1 - c_2 + e_2}{c_1 - c_2 + k_2}$ és $\frac{e_1}{k_1} < \frac{c_1 - c_3 + e_3}{c_1 - c_3 + k_3}$, akkor a $p = \left(1 - \frac{e_1}{k_1}, 0, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,i\,i$) Ha $\frac{e_2}{k_2} < 1$, $\frac{e_2}{k_2} < \frac{c_2 - c_1 + e_1}{c_2 - c_1 + k_1}$ és $\frac{e_2}{k_2} < \frac{c_2 - c_3 + e_3}{c_2 - c_3 + k_3}$, akkor a $p = \left(0, 1 - \frac{e_2}{k_2}, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

($i\,v$) Ha $\frac{e_3}{k_3} < 1$, $\frac{e_3}{k_3} < \frac{c_3 - c_1 + e_1}{c_3 - c_1 + k_1}$ és $\frac{e_3}{k_3} < \frac{c_3 - c_2 + e_2}{c_3 - c_2 + k_2}$, akkor a $p = \left(0, 0, 1 - \frac{e_3}{k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

BIZONYÍTÁS

A következő egyenlőtlenségeknek kell teljesülniük:

(i) $\quad -e_i + k_i < 0 \Leftrightarrow k_i < e_i \Leftrightarrow 1 < \frac{e_i}{k_i}$, $i = 1, 2, 3$

($i\,i$) $\quad e_1 - k_1 < 0 \Leftrightarrow e_1 < k_1 \Leftrightarrow \frac{e_1}{k_1} < 1$ és

$$\frac{c_1 e_1 - c_2 e_1 - c_1 k_1 + c_2 k_1 - e_2 k_1 + e_1 k_2}{k_1} < 0 \Leftrightarrow c_1 e_1 - c_2 e_1 + e_1 k_2 < c_1 k_1 - c_2 k_1 + e_2 k_1 \Leftrightarrow$$

$$\Leftrightarrow e_1(c_1 - c_2 + k_2) < k_1(c_1 - c_2 + e_2) \Leftrightarrow \frac{e_1}{k_1} < \frac{c_1 - c_2 + e_2}{c_1 - c_2 + k_2} \text{ és}$$

$$\frac{c_1 e_1 - c_3 e_1 - c_1 k_1 + c_3 k_1 - e_3 k_1 + e_1 k_3}{k_1} < 0 \Leftrightarrow c_1 e_1 - c_3 e_1 + e_1 k_3 < c_1 k_1 - c_3 k_1 + e_3 k_1 \Leftrightarrow$$

$$\Leftrightarrow e_1(c_1 - c_3 + k_3) < k_1(c_1 - c_3 + e_3) \Leftrightarrow \frac{e_1}{k_1} < \frac{c_1 - c_3 + e_3}{c_1 - c_3 + k_3}$$

($i\,i\,i$) és ($i\,v$) bizonyítása az előző ($i\,i$) bizonyítás analógiája, mivel a modell szimmetrikus.

■

Két faj együttélése esetén a sajátértékek valós részének negativitását feltételeit nehéz lenne belátni, ezért alkalmazzuk Michael Y. Li tételét $n = 3$ esetén:

$$A^{[2]} = \begin{pmatrix} a_{11} + a_{22} & a_{23} & -a_{13} \\ a_{32} & a_{11} + a_{33} & a_{12} \\ -a_{31} & a_{21} & a_{22} + a_{33} \end{pmatrix}. \tag{5.3.4}$$

Helyettesítsük be a $\left(\frac{e_2(c_2+k_1)-(c_2+e_1)k_2+c_1(-e_2+k_2)}{(c_1-c_2)(c_1-c_2-k_1+k_2)}, \frac{c_1(e_1-k_1)-e_2 k_1+c_2(-e_1+k_1)+e_1 k_2}{(c_1-c_2)(c_1-c_2-k_1+k_2)}, 0\right)$ egyensúlyi helyzetet a megfelelő Jacobi mátrixba, melynek determinánsa:

```
Jacobi03 =
  FullSimplify[(SzimFel31Jacobi /. SzimFel31Egyensuly[[3]])];
Jacobi03 // MatrixForm
FullSimplify[Det[Jacobi03]]
```

$$((c2\ (e1 - k1) + e2\ k1 + c1\ (-e1 + k1) - e1\ k2)\ (e2\ (c2 + k1) - (c2 + e1)\ k2 + c1\ (-e2 + k2))\ (-e3\ k1 + c3\ (-e1 + e2 + k1 - k2) - e1\ k2 + e3\ k2 + e2\ (k1 - k3) + c1\ (-e2 + e3 + k2 - k3) + e1\ k3 + c2\ (e1 - e3 - k1 + k3)))\Big/\left((c1 - c2)\ (c1 - c2 - k1 + k2)^2\right)$$

$A^{[2]}$:

```
CompM[M_ /; Dimensions[M] === {3, 3}] :=
 ( M[[1, 1]] + M[[2, 2]]    M[[2, 3]]               -M[[1, 3]]
   M[[3, 2]]                M[[1, 1]] + M[[3, 3]]   M[[1, 2]]
   -M[[3, 1]]               M[[2, 1]]               M[[2, 2]] + M[[3, 3]] )
CompM[Jacobi03] // MatrixForm
```

Még ennek sajátértékeit is nehéz megállapítani, így csupán szükséges, de nem elegendő feltételt adunk meg azon egyensúlyi helyzetekre, amikor csak az egyik faj hal ki.

5.3.2. TÉTEL

(*i*) Az $\left(\frac{e_2(c_2+k_1)-(c_2+e_1)k_2+c_1(-e_2+k_2)}{(c_1-c_2)(c_1-c_2-k_1+k_2)}, \frac{c_1(e_1-k_1)-e_2 k_1+c_2(-e_1+k_1)+e_1 k_2}{(c_1-c_2)(c_1-c_2-k_1+k_2)}, 0\right)$ egyensúlyi helyzet aszimptotikus stabilitásának szükséges feltétele: $c_1 - c_2 > k_1 - k_2$ esetén $k_1 - k_2 \le e_1 - e_2 \le c_1 - c_2$, $c_1 - c_2 < k_1 - k_2$ esetén $e_1 - e_2 \le k_1 - k_2$ és $e_1 - e_2 \le c_1 - c_2$.

(*i i*) Az $\left(\frac{e_3(c_3+k_1)-(c_3+e_1)k_3+c_1(-e_3+k_3)}{(c_1-c_3)(c_1-c_3-k_1+k_3)}, 0, \frac{c_1(e_1-k_1)-e_3 k_1+c_3(-e_1+k_1)+e_1 k_3}{(c_1-c_3)(c_1-c_3-k_1+k_3)}\right)$ egyensúlyi helyzet aszimptotikus stabilitásának szükséges feltétele: $c_1 - c_3 > k_1 - k_3$ esetén $k_1 - k_3 \le e_1 - e_3 \le c_1 - c_3$, $c_1 - c_3 < k_1 - k_3$ esetén $e_1 - e_3 \le k_1 - k_3$ és $e_1 - e_3 \le c_1 - c_3$.

(*i i i*) A $\left(0, \frac{e_3(c_3+k_2)-(c_3+e_2)k_3+c_2(-e_3+k_3)}{(c_2-c_3)(c_2-c_3-k_2+k_3)}, \frac{c_2(e_2-k_2)-e_3 k_2+c_3(-e_2+k_2)+e_2 k_3}{(c_2-c_3)(c_2-c_3-k_2+k_3)}\right)$ egyensúlyi helyzet aszimptotikus stabilitásának szükséges feltétele: $c_2 - c_3 > k_2 - k_3$ esetén $k_2 - k_3 \le e_2 - e_3 \le c_2 - c_3$, $c_2 - c_3 < k_2 - k_3$ esetén $e_2 - e_3 \le k_2 - k_3$ és $e_2 - e_3 \le c_2 - c_3$.

BIZONYÍTÁS

(*i*) A $0 \le p_1(\mathrm{t})+p_2(\mathrm{t})+p_3(t) \le 1$ feltételt vizsgáljuk:

```
FullSimplify[(e2 (c2 + k1) - (c2 + e1) k2 + c1 (-e2 + k2))/((c1 - c2) (c1 - c2 - k1 + k2)) +
  (c1 (e1 - k1) - e2 k1 + c2 (-e1 + k1) + e1 k2)/((c1 - c2) (c1 - c2 - k1 + k2))]
(e1 - e2 - k1 + k2)/(c1 - c2 - k1 + k2)
```

$p_1(\mathrm{t})+p_2(\mathrm{t})+p_3(t) = \frac{e_1-e_2-k_1+k_2}{c_1-c_2-k_1+k_2}$. Nézzük meg, mikor teljesülnek a feltételek:

$\frac{e_1-e_2-k_1+k_2}{c_1-c_2-k_1+k_2} \leq 1 \Leftrightarrow e_1 - e_2 - k_1 + k_2 \leq c_1 - c_2 - k_1 + k_2 \Leftrightarrow e_1 - e_2 \leq c_1 - c_2$, és

$\frac{e_1-e_2-k_1+k_2}{c_1-c_2-k_1+k_2} = \frac{(e_1-e_2)-(k_1-k_2)}{(c_1-c_2)-(k_1-k_2)} \geq 0$

□ *1. eset:* $c_1 - c_2 > k_1 - k_2$

$\frac{(e_1-e_2)-(k_1-k_2)}{(c_1-c_2)-(k_1-k_2)} \geq 0 \Leftrightarrow (e_1 - e_2) - (k_1 - k_2) \geq 0 \Leftrightarrow e_1 - e_2 \geq k_1 - k_2$.

□ *2. eset:* $c_1 - c_2 < k_1 - k_2$

$\frac{(e_1-e_2)-(k_1-k_2)}{(c_1-c_2)-(k_1-k_2)} \geq 0 \Leftrightarrow (e_1 - e_2) - (k_1 - k_2) \leq 0 \Leftrightarrow e_1 - e_2 \leq k_1 - k_2$.

($i\,i$) és ($i\,i\,i$) bizonyítása a fenti (i) bizonyításának analógiája.

■

5.3.4. Grafikus megjelenítés - interaktív vizsgálatok

Lehetőség van a rendszer interaktív vizsgálatára. Kézzel változtathatjuk a paraméterek értékeit, és láthatjuk hogyan változik az iránymező és az adott kezdeti értékekhez tartozó megoldás.



5.3.2. MEGJEGYZÉS

A kétfajos szimmetrikus felülkolonizációs modellben látottakhoz hasonlóan végezzük el a $c_{12} = c_1 - c_2, c_{13} = c_1 - c_3, c_{23} = c_2 - c_3$ helyettesítést, és vizsgáljuk meg a modellt.

Ekkor az új modell alakja a következő:

$$\begin{aligned} p_1'(t) &= k_1\, p_1(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) + \\ &\quad c_{12}\, p_1(t)\, p_2(t) + c_{13}\, p_1(t)\, p_3(t) - e_1\, p_1(t) \\ p_2'(t) &= k_2\, p_2(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) - \\ &\quad c_{12}\, p_1(t)\, p_2(t) + c_{23}\, p_2(t)\, p_3(t) - e_2\, p_2(t) \\ p_3'(t) &= k_3\, p_3(t)\,(1 - p_1(t) - p_2(t) - p_3(t)) - \\ &\quad c_{13}\, p_1(t)\, p_3(t) - c_{23}\, p_2(t)\, p_3(t) - e_3\, p_3(t). \end{aligned} \tag{5.3.5}$$

Nézzük meg az (5.3.5) egyenletrendszer egyensúlyi helyzeteit!

Változók és paraméterek:

```
Clear[k1, k2, k3, e1, e2, e3, c1, c2, c3];
var = {p1, p2, p3};
 parm = {k1, k2, k3, e1, e2, e3, c1, c2, c3};
```

Az (5.3.5) egyenletrendszer jobb oldala:

```
SzimFel32 = {p1 (k1 (1 - p1 - p2 - p3) - e1 + p2 c12 + p3 c13),
p2 (k2 (1 - p1 - p2 - p3) - e2 - p1 c12 + p3 c23),
p3 (k3 (1 - p1 - p2 - p3) - e3 - p1 c13 - p2 c23)};
```

Az (5.3.5) egyenletrendszer egyensúlyi helyzetei:

```
SzimFel32Egyensuly =
  FullSimplify[Solve[SzimFel32 == {0, 0, 0}, var]];
```

```
SzimFel32Egyensuly // Column
```

$$\{p1 \to 0,\ p2 \to 0,\ p3 \to 0\}$$

$$\left\{p1 \to 1 - \frac{e1}{k1},\ p2 \to 0,\ p3 \to 0\right\}$$

$$\left\{p1 \to \frac{e2\ k1 - e1\ k2 + c12\ (-e2 + k2)}{c12\ (c12 - k1 + k2)},\ p2 \to \frac{c12\ (e1 - k1) - e2\ k1 + e1\ k2}{c12\ (c12 - k1 + k2)},\ p3 \to 0\right\}$$

$$\left\{p1 \to \frac{e3\ k1 - e1\ k3 + c13\ (-e3 + k3)}{c13\ (c13 - k1 + k3)},\ p3 \to \frac{c13\ (e1 - k1) - e3\ k1 + e1\ k3}{c13\ (c13 - k1 + k3)},\ p2 \to 0\right\}$$

$$\Big\{p1 \to \big(c23^2\ (-e1 + k1) + (c12 - c13)\ (e3\ k2 - e2\ k3) + c23\ (-e2\ k1 + e3\ k1 + c13\ (e2 - k2) + e1\ k2 - e1\ k3 + c12\ (-e3 + k3))\big) \Big/ ((c12 - c13 + c23)\ (c23\ k1 - c13\ k2 + c12\ k3)),$$
$$p2 \to \frac{c13^2\ (-e2 + k2) - (c12 + c23)\ (e3\ k1 - e1\ k3) + c13\ (c23\ (e1 - k1) + e2\ k1 - e1\ k2 + e3\ (c12 + k2) - (c12 + e2)\ k3)}{(c12 - c13 + c23)\ (c23\ k1 - c13\ k2 + c12\ k3)}$$
$$,\ p3 \to \big(-(c13 - c23)\ (e2\ k1 - e1\ k2) + c12^2\ (-e3 + k3) + c12\ (c23\ (-e1 + k1) + c13\ (e2 - k2) + e3\ (k1 - k2) + (-e1 + e2)\ k3)\big) \Big/ ((c12 - c13 + c23)\ (c23\ k1 - c13\ k2 + c12\ k3))\Big\}$$

$$\left\{p2 \to 1 - \frac{e2}{k2},\ p1 \to 0,\ p3 \to 0\right\}$$

$$\left\{p2 \to \frac{e3\ k2 - e2\ k3 + c23\ (-e3 + k3)}{c23\ (c23 - k2 + k3)},\ p3 \to \frac{c23\ (e2 - k2) - e3\ k2 + e2\ k3}{c23\ (c23 - k2 + k3)},\ p1 \to 0\right\}$$

$$\left\{p3 \to 1 - \frac{e3}{k3},\ p1 \to 0,\ p2 \to 0\right\}$$

Láthatjuk, hogy ebben az esetben van együttélés. A folyamatábra segítségével egyértelműen látható, hogy elvárásainkkal és számításainkkal ellentétben ez a modell a hierarchikus modellel ekvivalens.



5.3.2. ábra

Általános felülkolonizációs modell három fajra; e_i a kihalási ráta, k_i a kolonizációs ráta, p_i az elfoglalt folt, c_{ij} a felülkolonizációs ráták különbsége, L a lakatlan (üres) folt ($i, j = 1, 2, 3;\ i < j$)

5.4. Szimmetrikus felülkolonizációs modell megszorításokkal

Tegyük meg a következő megszorításokat: egyezzen mindegyik fajnál a kolonizációs ráta a felülkolonizációs rátával, azaz legyen:

$$c_1 = k_1,\ c_2 = k_2,\ c_3 = k_3.$$

```
FullSimplify[
 {p1 (k1 (1 - p1 - p2 - p3) - e1 + p2 (c1 - c2) + p3 (c1 - c3)),

   p2 (k2 (1 - p1 - p2 - p3) - e2 + p1 (-c1 + c2) + p3 (c2 - c3)),
            p3 (k3 (1 - p1 - p2 - p3) - e3 + p1 (-c1 + c3) +
     p2 (-c2 + c3))} /. {c1 → k1, c2 → k2, c3 → k3}]
```

```
{-p1 (e1 + k1 (-1 + p1) + k2 p2 + k3 p3),
 -p2 (e2 + k1 p1 + k2 (-1 + p2) + k3 p3),
 -(e3 + k1 p1 + k2 p2 + k3 (-1 + p3)) p3}
```

5.4.1. A modell

Egy faj kolonizációjának sebessége arányos az elfoglalt és az el nem foglalt területek arányának szorzatával. A halálozás csak az elfoglalt foltok arányával arányos (ez a faj belső jellemzője). Egy faj felülkolonizációjának sebessége egyenesen arányos az általa elfoglalt és egy másik faj által elfoglalt terület arányának szorzatával. Legyen $p_1(t)$ az első, $p_2(t)$ a második, $p_3(t)$ a harmadik ($0 \leq p_1(t) + p_2(t) + p_3(t) \leq 1$) faj által elfoglalt terület aránya a t időpillanatban. Megszorításokkal a háromfajos felülkolonizációs modell az alábbi differenciálegyenlet rendszerrel definiálható:

$$\begin{aligned} p_1'(t) &= k_1\, p_1(t)\,(1 - p_1(t)) - k_2\, p_1(t)\, p_2(t) - k_3\, p_1(t)\, p_3(t) - e_1\, p_1(t) \\ p_2'(t) &= k_2\, p_2(t)\,(1 - p_2(t)) - k_1\, p_1(t)\, p_2(t) - k_3\, p_2(t)\, p_3(t) - e_2\, p_2(t) \\ p_3'(t) &= k_3\, p_3(t)\,(1 - p_3(t)) - k_1\, p_1(t)\, p_3(t) - k_2\, p_2(t)\, p_3(t) - e_3\, p_3(t) \end{aligned} \qquad (5.4.6)$$

ahol k_1, k_2, k_3 ($0 \leq k_i,\ i = 1, 2, 3$) konstansok a globális kolonizációs és felülkolonizációs ráták is egyben, e_1, e_2, e_3 ($0 \leq e_i,\ i = 1, 2, 3$) konstansok a globális halálozási ráták.

A továbbiakban az egyszerűség kedvéért használjuk a következő jelölést: $p_i := p_i(t),\ i = 1, 2, 3$. Az alábbi ábra mutatja az egyes fajok által elfoglalt és üres foltok közti változást.



5.4.1. ábra

Szimmetrikus felülkolonizációs modell három fajra – megszorításokkal; e_i a kihalási ráta, k_i a kolonizációs ráta, p_i az elfoglalt folt, L a lakatlan (üres) folt ($i = 1,\ 2,\ 3$)

5.4.2. Az egyensúlyi helyzetek

Változók és paraméterek:

```
Clear[k1, k2, k3, e1, e2, e3, c1, c2, c3];
var = {p1, p2, p3};
parm = {k1, k2, k3, e1, e2, e3};
```

Az (5.4.6) egyenletrendszer jobb oldala:

```
SzimFel33 = {p1 (k1 (1 - p1) - e1 - k2 p2 - k3 p3 ),
             p2 (k2 (1 - p2) - e2 - k1 p1 - k3 p3 ),
             p3 (k3 (1 - p3) - e3 - k1 p1  - k2 p2 ) };
```

Az (5.4.6) egyenletrendszer egyensúlyi helyzetei:

```
SzimFel33Egyensuly =
  FullSimplify[Solve[SzimFel33 == {0, 0, 0}, var]];
SzimFel33Egyensuly // Column
```

$$\{p1 \to 0,\ p2 \to 0,\ p3 \to 0\}$$
$$\left\{p1 \to 1 - \frac{e1}{k1},\ p2 \to 0,\ p3 \to 0\right\}$$
$$\left\{p2 \to 1 - \frac{e2}{k2},\ p1 \to 0,\ p3 \to 0\right\}$$
$$\left\{p3 \to 1 - \frac{e3}{k3},\ p1 \to 0,\ p2 \to 0\right\}$$

Ugyanazon egyensúlyi helyzeteket kaptuk, mint a háromfajos preemptív modell esetében. Így ennek a rendszernek sincs olyan egyensúlyi helyzete, ahol legalább két faj által elfoglalt terület pozitív. Szükséges az $\frac{e_i}{k_i} \le 1,\ i = 1, 2, 3$ feltétel teljesülése (lásd az (5.1.1) megjegyzést).

5.4.3. Az egyensúlyi helyzetek stabilitási tulajdonságai

Az (5.4.6) egyenletrendszer jobb oldalának Jacobi mátrixai:

```
SzimFel33Jacobi = FullSimplify[D[SzimFel33, {var}]]
```

```
{{-e1 + k1 - 2 k1 p1 - k2 p2 - k3 p3, -k2 p1, -k3 p1},
 {-k1 p2, -e2 + k2 - k1 p1 - 2 k2 p2 - k3 p3, -k3 p2},
 {-k1 p3, -k2 p3, -e3 + k3 - k1 p1 - k2 p2 - 2 k3 p3}}
```

```
linSzimFel33 = SzimFel33Jacobi /. SzimFel33Egyensuly;
Map[MatrixForm, linSzimFel33] // Column
```

$$\begin{pmatrix} -e1+k1 & 0 & 0 \\ 0 & -e2+k2 & 0 \\ 0 & 0 & -e3+k3 \end{pmatrix}$$

$$\begin{pmatrix} -e1+k1-2\left(1-\frac{e1}{k1}\right)k1 & -\left(1-\frac{e1}{k1}\right)k2 & -\left(1-\frac{e1}{k1}\right)k3 \\ 0 & -e2-\left(1-\frac{e1}{k1}\right)k1+k2 & 0 \\ 0 & 0 & -e3-\left(1-\frac{e1}{k1}\right)k1+k3 \end{pmatrix}$$

$$\begin{pmatrix} -e1+k1-\left(1-\frac{e2}{k2}\right)k2 & 0 & 0 \\ -k1\left(1-\frac{e2}{k2}\right) & -e2+k2-2\left(1-\frac{e2}{k2}\right)k2 & -\left(1-\frac{e2}{k2}\right)k3 \\ 0 & 0 & -e3-\left(1-\frac{e2}{k2}\right)k2+k3 \end{pmatrix}$$

$$\begin{pmatrix} -e1+k1-\left(1-\frac{e3}{k3}\right)k3 & 0 & 0 \\ 0 & -e2+k2-\left(1-\frac{e3}{k3}\right)k3 & 0 \\ -k1\left(1-\frac{e3}{k3}\right) & -k2\left(1-\frac{e3}{k3}\right) & -e3+k3-2\left(1-\frac{e3}{k3}\right)k3 \end{pmatrix}$$

A Jacobi mátrixok sajátértékei rendre:

```
ev = Map[MatrixForm,
  Map[FullSimplify[Eigenvalues[#]] &, linSzimFel33]]
```

$$\left\{\begin{pmatrix} -e1+k1 \\ -e2+k2 \\ -e3+k3 \end{pmatrix}, \begin{pmatrix} e1-k1 \\ e1-e2-k1+k2 \\ e1-e3-k1+k3 \end{pmatrix}, \begin{pmatrix} e2-k2 \\ -e1+e2+k1-k2 \\ e2-e3-k2+k3 \end{pmatrix}, \begin{pmatrix} e3-k3 \\ -e1+e3+k1-k3 \\ -e2+e3+k2-k3 \end{pmatrix}\right\}$$

5.4.1. TÉTEL

(*i*) Ha $1 < \frac{e_i}{k_i}$, $i = 1, 2, 3$, akkor a $p = (0, 0, 0)$ egyensúlyi helyzet aszimptotikusan stabilis.

(*i i*) Ha $\frac{e_1}{k_1} < 1$, $e_1 - k_1 < e_2 - k_2$ és $e_1 - k_1 < e_3 - k_3$, akkor a $p = \left(1 - \frac{e_1}{k_1}, 0, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

(*i i i*) Ha $\frac{e_2}{k_2} < 1$, $e_2 - k_2 < e_1 - k_1$ és $e_2 - k_2 < e_3 - k_3$, akkor a $p = \left(0, 1 - \frac{e_2}{k_2}, 0\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

(*i v*) Ha $\frac{e_3}{k_3} < 1$, $e_3 - k_3 < e_1 - k_1$ és $e_3 - k_3 < e_2 - k_2$, akkor a $p = \left(0, 0, 1 - \frac{e_3}{k_3}\right)$ egyensúlyi helyzet aszimptotikusan stabilis.

BIZONYÍTÁS

A következő egyenlőtlenségeknek kell teljesülniük:

(i) $\quad -e_i + k_i < 0 \iff k_i < e_i \iff 1 < \frac{e_i}{k_i},\ i = 1, 2$

$(i\,i)$ $\quad e_1 - k_1 < 0 \Leftrightarrow e_1 < k_1 \iff \frac{e_1}{k_1} < 1$ és

$\quad e_1 - e_j - k_1 + k_j < 0 \Leftrightarrow e_1 - k_1 < e_j - k_j,\ j = 2, 3$

$(i\,i\,i)$ és $(i\,v)$ az előző $(i\,i)$ bizonyításának analjógiája.

■

5.4.4. Grafikus megjelenítés - interaktív vizsgálatok

Lehetőség van a rendszer interaktív vizsgálatára. Kézzel változtathatjuk a paraméterek értékeit, és láthatjuk hogyan változik az iránymező és az adott kezdeti értékekhez tartozó megoldás.



Az eredmények összefoglalása

Az összes általunk vizsgált három fajosmodellt tekintve, kissé meglepő módon csak felülkolonizációs esetben tapasztalhatjuk a hármas együttélést, azaz csak akkor, ha van kölcsönhatás a fajok között. Mivel hat vagy akár kilenc paraméterrel illetve három változóval így három egyenletből álló differenciálegyenlet rendszerrel van dolgunk, ezért számítottunk is arra, hogy sok esetben nehéz feladat a szükséges és elegendő feltétel meghatározása az aszimptotikus egyensúlyi helyzet létezésére.